

NAG Library Routine Document

X04CDF

Note: before using this routine, please read the Users' Note for your implementation to check the interpretation of ***bold italicised*** terms and other implementation-dependent details.

1 Purpose

X04CDF prints a ***double precision*** triangular matrix stored in a packed one-dimensional array.

2 Specification

```

SUBROUTINE X04CDF(UPLO, DIAG, N, A, FORM, TITLE, LABROW, RLABS, LABCOL,
1              CLABS, NCOLS, INDENT, IFAIL)
INTEGER          N, NCOLS, INDENT, IFAIL
double precision A(*)
CHARACTER*(*)    FORM, TITLE, RLABS(*), CLABS(*)
CHARACTER*1      UPLO, DIAG, LABROW, LABCOL

```

3 Description

X04CDF prints a ***double precision*** triangular matrix stored in packed form, using a format specifier supplied by you. The matrix must be packed by column. The matrix is output to the unit defined by X04ABF.

4 References

None.

5 Parameters

- 1: UPLO – CHARACTER*1 *Input*
On entry: indicates the type of the matrix to be printed
UPLO = 'L'
The matrix is lower triangular. In this case, the packed array A holds the matrix elements in the following order: (1, 1), (2, 1), ..., (N, 1), (2, 2), (3, 2), ..., (N, 2), etc..
UPLO = 'U'
The matrix is upper triangular. In this case, the packed array A holds the matrix elements in the following order: (1, 1), (1, 2), (2, 2), (1, 3), (2, 3), (3, 3), (1, 4), etc..
Constraint: UPLO = 'L' or 'U'.
- 2: DIAG – CHARACTER*1 *Input*
On entry: indicates whether the diagonal elements of the matrix are to be printed.
DIAG = 'B'
The diagonal elements of the matrix are not referenced and not printed.
DIAG = 'U'
The diagonal elements of the matrix are not referenced, but are assumed all to be unity, and are printed as such.
DIAG = 'N'
The diagonal elements of the matrix are referenced and printed.
Constraint: DIAG = 'B', 'U' or 'N'.

- 3: N – INTEGER *Input*
On entry: the order of the matrix to be printed.
 If N is less than 1, X04CDF will exit immediately after printing TITLE; no row or column labels are printed.
- 4: A(*) – **double precision** array *Input*
Note: the dimension of the array A must be at least $\max(1, N \times (N + 1)/2)$.
On entry: the matrix to be printed. Note that A must have space for the diagonal elements of the matrix, even if these are not stored.
 More precisely,
 if UPLO = 'U', the upper triangle of A must be stored with element A_{ij} in $A(i + j(j - 1)/2)$ for $i \leq j$;
 if UPLO = 'L', the lower triangle of A must be stored with element A_{ij} in $A(i + (2n - j)(j - 1)/2)$ for $i \geq j$.
 If DIAG = 'U', the diagonal elements of A are assumed to be 1, and are not referenced; the same storage scheme is used whether DIAG = 'N' or 'U'.
- 5: FORM – CHARACTER*(*) *Input*
On entry: a valid Fortran format code. This may be any format code allowed on the system, whether it is standard Fortran or not. FORM is used to print elements of the matrix A. It may or may not be enclosed in brackets. Examples of valid values for FORM are 'F11.4', '1PE13.5', 'G14.5'.
 In addition, there are the following special codes which force X04CDF to choose its own format code:
 FORM = ' '
 X04CDF will choose a format code such that numbers will be printed with either an F8.4, an F11.4 or a 1PE13.4 format. The F8.4 code is chosen if the sizes of all the matrix elements to be printed lie between 0.001 and 1.0. The F11.4 code is chosen if the sizes of all the matrix elements to be printed lie between 0.001 and 9999.9999. Otherwise the 1PE13.4 code is chosen.
 FORM = ' * '
 X04CDF will choose a format code such that numbers will be printed to as many significant digits as are necessary to distinguish between neighbouring machine numbers. Thus any two numbers that are stored with different internal representations should look different on output. Whether they do in fact look different will depend on the run-time library of the Fortran compiler in use.
 FORM = 'MATLAB' followed by any of the above, e.g., FORM = 'MATLAB F11.4',
 FORM = 'MATLAB * '
 X04CDF will print the matrix such that it can be input into MATLAB. Elements of the matrix will be printed with format specified by the format code following 'MATLAB'. TITLE will be used as the name of the matrix.
Constraint: the character length of FORM must be ≤ 80 .
- 6: TITLE – CHARACTER*(*) *Input*
On entry: a title to be printed above the matrix or name of the matrix.
 If TITLE = ' ', no title (and no blank line) will be printed.
 If TITLE contains more than NCOLS characters, the contents of TITLE will be wrapped onto more than one line, with the break after NCOLS characters.
 Any trailing blank characters in TITLE are ignored.

If FORM = 'MATLAB', TITLE will be used as the name of the matrix.

- 7: LABROW – CHARACTER*1 *Input*
On entry: indicates the type of labelling to be applied to the rows of the matrix.
 LABROW = 'N'
 Prints no row labels.
 LABROW = 'I'
 Prints integer row labels.
 LABROW = 'C'
 Prints character labels, which must be supplied in array RLABS.
Constraint: LABROW = 'N', 'I' or 'C'.
- 8: RLABS(*) – CHARACTER*(*) array *Input*
Note: the dimension of the array RLABS must be at least N if LABROW = 'C', and at least 1 otherwise.
On entry: if LABROW = 'C', must contain labels for the rows of the matrix.
 Labels are right-justified when output, in a field which is as wide as necessary to hold the longest row label. Note that this field width is subtracted from the number of usable columns, NCOLS.
- 9: LABCOL – CHARACTER*1 *Input*
On entry: indicates the type of labelling to be applied to the columns of the matrix.
 LABCOL = 'N'
 Prints no column labels.
 LABCOL = 'I'
 Prints integer column labels.
 LABCOL = 'C'
 Prints character labels, which must be supplied in array CLABS.
Constraint: LABCOL = 'N', 'I' or 'C'.
- 10: CLABS(*) – CHARACTER*(*) array *Input*
Note: the dimension of the array CLABS must be at least N if LABCOL = 'C', and at least 1 otherwise.
On entry: if LABCOL = 'C', must contain labels for the columns of the matrix.
 Labels are right-justified when output. Any label that is too long for the column width, which is determined by FORM, is truncated.
- 11: NCOLS – INTEGER *Input*
On entry: the maximum output record length. If the number of columns of the matrix is too large to be accommodated in NCOLS characters, the matrix will be printed in parts, containing the largest possible number of matrix columns, and each part separated by a blank line.
 NCOLS must be large enough to hold at least one column of the matrix using the format specifier in FORM. If a value less than 0 or greater than 132 is supplied for NCOLS, then the value 80 is used instead.
- 12: INDENT – INTEGER *Input*
On entry: the number of columns by which the matrix (and any title and labels) should be indented. The effective value of NCOLS is reduced by INDENT columns. If a value less than 0 or greater than NCOLS is supplied for INDENT, the value 0 is used instead.

13: IFAIL – INTEGER

Input/Output

On entry: IFAIL must be set to 0, -1 or 1. If you are unfamiliar with this parameter you should refer to Section 3.3 in the Essential Introduction for details.

On exit: IFAIL = 0 unless the routine detects an error (see Section 6).

For environments where it might be inappropriate to halt program execution when an error is detected, the value -1 or 1 is recommended. If the output of error messages is undesirable, then the value 1 is recommended. Otherwise, if you are not familiar with this parameter, the recommended value is 0. **When the value -1 or 1 is used it is essential to test the value of IFAIL on exit.**

6 Error Indicators and Warnings

If on entry IFAIL = 0 or -1, explanatory error messages are output on the current error message unit (as defined by X04AAF).

Errors or warnings detected by the routine:

IFAIL = 1

On entry, UPLO $\neq$ 'L' or 'U'.

IFAIL = 2

On entry, DIAG $\neq$ 'N', 'U' or 'B'.

IFAIL = 3

On entry, variable FORM is more than 80 characters long.

IFAIL = 4

The code supplied in FORM cannot be used to output a number. FORM probably has too wide a field width or contains an illegal edit descriptor.

IFAIL = 5

On entry, either LABROW or LABCOL $\neq$ 'N', 'I' or 'C'.

IFAIL = 6

The quantity NCOLS – INDENT – *labwid* (where *labwid* is the width needed for the row labels) is not large enough to hold at least one column of the matrix.

7 Accuracy

Not applicable.

8 Further Comments

None.

9 Example

This example calls X04CDF three times, first to print 4 by 4 lower triangular matrix, and then twice to print a 5 by 5 upper triangular matrix; various options for labelling and formatting are illustrated.

9.1 Program Text

```

*      X04CDF Example Program Text
*      Mark 14 Release. NAG Copyright 1989.
*      .. Parameters ..
      INTEGER          NOUT
      PARAMETER        (NOUT=6)
      INTEGER          NMAX, LA
      PARAMETER        (NMAX=5,LA=(NMAX*(NMAX+1))/2)
*      .. Local Scalars ..
      INTEGER          I, IFAIL, INDENT, NCOLS
*      .. Local Arrays ..
      DOUBLE PRECISION A(LA)
      CHARACTER*7      CLABS(NMAX), RLABS(NMAX)
*      .. External Subroutines ..
      EXTERNAL         X04CDF
*      .. Data statements ..
      DATA            CLABS/'Un', 'Deux', 'Trois', 'Quatre', 'Cinq'/
      DATA            RLABS/'Uno', 'Duo', 'Tre', 'Quattro', 'Cinque'/
*      .. Executable Statements ..
      WRITE (NOUT,*) 'X04CDF Example Program Results'
      WRITE (NOUT,*)

*
*      Generate an array of data
      DO 20 I = 1, LA
        A(I) = I
20 CONTINUE
*
      NCOLS = 80
      INDENT = 0
      IFAIL = 0
*
*      Print order 4 lower triangular matrix with default format and
*      integer row and column labels
      CALL X04CDF('Lower','Non-unit',4,A,' ','Example 1:','Integer',
+              RLABS,'Integer',CLABS,NCOLS,INDENT,IFAIL)
*
      WRITE (NOUT,*)

*
*      Print order 5 upper triangular matrix with user-supplied format
*      and row and column labels
      CALL X04CDF('Upper','Unit',5,A,'F8.2','Example 2:','Character',
+              RLABS,'Character',CLABS,NCOLS,INDENT,IFAIL)
*
      END

```

9.2 Program Data

None.

9.3 Program Results

X04CDF Example Program Results

Example 1:

	1	2	3	4
1	1.0000			
2	2.0000	5.0000		
3	3.0000	6.0000	8.0000	
4	4.0000	7.0000	9.0000	10.0000

Example 2:

	Un	Deux	Trois	Quatre	Cinq
Uno	1.00	2.00	4.00	7.00	11.00
Duo		1.00	5.00	8.00	12.00
Tre			1.00	9.00	13.00
Quattro				1.00	14.00
Cinque					1.00